[bookmark: _GoBack]* RendallGreulich_SAScode.sas ;

options nocenter ;
libname silc 'C:\Users\mrendall\Documents\UMD\proposal\ComparativeFertility\SILC';

* data file PolandPartner2 input variables: dv, fulltimeEMPL1, fulltimeEMPL2, fulltime1P age.

The correspondence to variables in the article is as follows: 
  dv = Y(t)
  fulltimeEMPL1 = E(t-1)
  fulltimeEMPL2 = E(t-2)
  fulltime1P (partner fulltime employed at t-1) = Z(t-1)
  age = Z(t-1) ;

data Poland;
   set silc.PolandPartner2(where=(fulltimeEMPL1 in(0,1) and dv in(0,1) and fulltime1P in(0,1)));

 * transforming employment history to get Sample size column data for Table 1 
    and to construct regressors for Models 1 and 2 in Table 2. 

  The correspondence to variables in the article is as follows: 
    fulltimet1only = D1 
    fulltimet1andt2 = D2;

ft1ft2 = -9;
fulltimet1andt2=-9;
fulltimet1only=-9;
if fulltimeEMPL2 eq . and fulltimeEMPL1 eq 1 then 
do; 
   fulltimet1andt2=.;
   fulltimet1only=.;
   ft1ft2=-8;
end;
if fulltimeEMPL2 in(.,0,1) and fulltimeEMPL1 eq 0 then 
do;
   fulltimet1only=0;
   fulltimet1andt2=0;
   ft1ft2=0;
end;
if fulltimeEMPL2 eq 0 and fulltimeEMPL1 eq 1 then 
do;
   fulltimet1only=1;
   fulltimet1andt2=0;
   ft1ft2=1;
end;
if fulltimeEMPL2 eq 1 and fulltimeEMPL1 eq 1 then 
do;
   fulltimet1only=0;
   fulltimet1andt2=1;
   ft1ft2=2;
end;

if fulltimeEMPL2 eq . and fulltimeEMPL1 eq . then 
do;
   fulltimet1andt2=.;
   fulltimet1only=.;
   ft1ft2=-8;
end;
run;

* frequencies for Table 1;

proc freq data=Poland;
table ft1ft2/missing out=ft1ft2 outpct;  
title 'employment status at t-2 and t-1';
run;
proc export data=ft1ft2
outfile = 'C:\Users\mrendall\Documents\UMD\proposal\ComparativeFertility\SILC\poland.xls'
	replace;
	sheet='ft1ft2_freq';
run;

* Regression with complete data, using employment status only at t-1 (Table 2 Model 1);

ods output PARAMETERESTIMATES=gmparms;
proc logistic  data=Poland;
  model dv (event='1') = fulltimeEMPL1 age  fulltime1P;
   run;
proc export data = gmparms
outfile = 'C:\Users\mrendall\Documents\UMD\proposal\ComparativeFertility\SILC\poland.xls'
replace;
sheet='complete_ft_t1';
run;

* Regression with complete data, using employment status at t-1 and t-2 (Table 2 Model 2);

ods output PARAMETERESTIMATES=gmparms;
proc logistic  data=Poland;
  model dv (event='1') = fulltimet1andt2 fulltimet1only age  fulltime1P;
   run;
proc export data = gmparms
outfile = 'C:\Users\mrendall\Documents\UMD\proposal\ComparativeFertility\SILC\poland.xls'
replace;
sheet='complete_ft_t1t2';
run;

* PROC MI, to estimate equation (1) in article;

proc mi data=Poland seed=100090 out=miout nimpute=20; 
 var dv fulltimeEMPL1 fulltimeEMPL2 age  fulltime1P;
 class dv fulltimeEMPL1 fulltimeEMPL2 fulltime1P;
 monotone logistic (fulltimeEMPL2= fulltimeEMPL1 age  fulltime1P dv /descending details) ; 
run;

* transforming employment history to construct regressors D1 and D2 for Table 2 Model 3;
data miout; set miout;
fulltimet1andt2=0;
if fulltimeEMPL2 eq 1 and fulltimeEMPL1 eq 1 then fulltimet1andt2=1;
fulltimet1only=0;
if fulltimeEMPL2 eq 0 and fulltimeEMPL1 eq 1 then fulltimet1only=1;
run;

* Regression with imputed fulltime employment status in t-2 where missing (Table 2 Model 3);

ods output PARAMETERESTIMATES=gmparms;
proc logistic  data=miout ;
model dv (event='1') = fulltimet1only fulltimet1andt2 age  fulltime1P;
   by _Imputation_ ;
   run;

* PROC MIANALYZE ;
ods output ParameterEstimates = analyze_estimates;
proc mianalyze parms=gmparms ;
   modeleffects Intercept fulltimet1only fulltimet1andt2 age  fulltime1P;
run; 
proc print data=analyze_estimates;
title 'analyze_estimates';
run;
proc export data = analyze_estimates
outfile = 'C:\Users\mrendall\Documents\UMD\proposal\ComparativeFertility\SILC\poland.xls'
replace;
sheet='mi20fulltimet1t2';
run;

 * Alternate coding of PROC MI and PROC MIANALYZE using STATA’s mi 

    use poland, clear
    keep dv fulltimeEMPL1 fulltimeEMPL2 age  fulltime1P
     mi set mlong
     mi misstable summarize
     mi register imputed fulltimeEMPL2
     mi impute logit fulltimeEMPL2 = age  dv fulltimeEMPL1 fulltime1P, add(20) rseed(2232)
     mi passive: gen fulltimet1andt2=0
     mi passive: replace fulltimet1andt2=1 if fulltimeEMPL2 == 1 & fulltimeEMPL1 == 1
     mi passive: gen fulltimet1only=0
     mi passive: replace fulltimet1only=1 if fulltimeEMPL2 == 0 & fulltimeEMPL1 == 1
     mi estimate: logit dv fulltimet1andt2 fulltimet1only age  fulltime1P

